﻿Shader "BuiltIn/WebAnimMapLitShader"
{
    Properties
    {
        [HDR]_Color ("Color", Color) = (1,1,1,1)
        _Glossiness ("Smoothness", Range(0,1)) = 0.5
        _Metallic ("Metallic", Range(0,1)) = 0.0
        [HDR]_EmissionColor("Emission Color", Color) = (0,0,0)

        [NoScaleOffset]_MainTex ("Texture", 2D) = "white" {}
		[NoScaleOffset]_AnimMap ("AnimMap", 2D) ="white" {}
        [NoScaleOffset]_NormalMap("NormalMap", 2D) ="white" {}
        _AnimStart("AnimStart", Float) = 0
		_AnimLen("Anim Length", Float) = 0
    }
    SubShader
    {
        Tags { "RenderType"="Opaque" }
        LOD 200

        CGPROGRAM
        // Physically based Standard lighting model, and enable shadows on all light types
        #pragma surface surf Standard addshadow fullforwardshadows vertex:vert

        // Use shader model 3.0 target, to get nicer looking lighting
        #pragma target 3.0

        sampler2D _MainTex;
        sampler2D _NormalMap;
        sampler2D _AnimMap;
        float4 _AnimMap_TexelSize; // x == 1/width

        float _AnimStart;
        float _AnimLen;

        struct Input
        {
            float2 uv_MainTex;
        };

        void vert(inout appdata_full v, out Input i)
		{
			UNITY_INITIALIZE_OUTPUT(Input, i);

            float f = _Time.y / _AnimLen;
            f = f + _AnimStart / _AnimLen;

            float animMap_x = (v.color.r + 0.5) * _AnimMap_TexelSize.x;
            float animMap_y = f;

            float4 pos = tex2Dlod(_AnimMap, float4(animMap_x, animMap_y, 0, 0));
            v.vertex.xyz = pos.xyz;

            float3 normal = tex2Dlod(_NormalMap, float4(animMap_x, animMap_y, 0, 0)).rgb * 2 - 1;
            v.normal = normal;
        }

        half _Glossiness;
        half _Metallic;
        fixed4 _Color;
        half4 _EmissionColor;

        // Add instancing support for this shader. You need to check 'Enable Instancing' on materials that use the shader.
        // See https://docs.unity3d.com/Manual/GPUInstancing.html for more information about instancing.
        // #pragma instancing_options assumeuniformscaling
        UNITY_INSTANCING_BUFFER_START(Props)
            // put more per-instance properties here
        UNITY_INSTANCING_BUFFER_END(Props)

        void surf (Input IN, inout SurfaceOutputStandard o)
        {
            // Albedo comes from a texture tinted by color
            fixed4 c = tex2D (_MainTex, IN.uv_MainTex) * _Color;

            o.Albedo = c.rgb;
            // Metallic and smoothness come from slider variables
            o.Metallic = _Metallic;
            o.Smoothness = _Glossiness;
            o.Emission = _EmissionColor;
            o.Alpha = c.a;
        }
        ENDCG
    }
    FallBack "Diffuse"
}
